

// Viertel Kreis //

UM ViertelKreis
WIEDERHOLE 45 [VORWÄRTS 2 RECHTS 2]
ENDE

// Blumenblatt = 2 Viertel Kreis //
UM Blumenblatt
WIEDERHOLE 2 [ViertelKreis RECHTS 90]
ENDE

// Blume = 10 Blumenblatt //

UM Blume
WIEDERHOLE 10 [Blumenblatt RECHTS 360/10]
ENDE
// Pflanze = Blume + Stamm + Blumenblatt + Stamm //

UM Pflanze
FARBEIGEL 100 // ROT //

Blume
FARBEIGEL 200 // Grün //

RÜCKWÄRTS 130

Blumenblatt

RÜCKWÄRTS 70

ENDE

UM Blumen

BILD BESCHLEUNIGE

STIFTHOCH LINKS 90 VORWÄRTS 150

RECHTS 90 STIFTAB

WIEDERHOLE 3 [

Pflanze STIFTHOCH VORWÄRTS 200 RECHTS 90 VORWÄRTS 150 LINKS 90 STIFTAB]

ENDE

// Tippen in die Befehllinie (Arbeitsfenster):

Blumen

//

UM Pappeln

BILD VERSTECKTIGEL BESCHLEUNIGE

SEGMENT [-20 0] [-20 25] Baum 40 0 (-20) 25

SEGMENT [-60 0] [-60 25] Baum 40 0 (-60) 25

SEGMENT [-100 0] [-100 25] Baum 40 0 (-100) 25

SEGMENT [-140 0] [-140 25] Baum 40 0 (-140) 25

SEGMENT [20 0] [20 25] Baum 40 0 20 25

SEGMENT [60 0] [60 25] Baum 40 0 60 25

SEGMENT [100 0] [100 25] Baum 40 0 100 25

SEGMENT [140 0] [140 25] Baum 40 0 140 25

ENDE

UM Baum :lZweig :cap :x :y

PRÜFE :lZweig > 3

WENNWAHR [

STIFTHOCH AUFXY :x :y STIFTAB

STARTEZUFALL 0 20

AUFKAP (:cap - 15) + (ZUFALLZAHL - 10)

VORWÄRTS :lZweig

Baum ((:lZweig * 1) / 3) KAP XKO YKO

STIFTHOCH AUFXY :x :y STIFTAB

STARTEZUFALL 0 20

AUFKAP :cap + (ZUFALLZAHL - 10)

VORWÄRTS :lZweig

Baum ((:lZweig * 4) / 5) KAP XKO YKO

STIFTHOCH AUFXY :x :y STIFTAB

STARTEZUFALL 0 20

AUFKAP (:cap + 15) + (ZUFALLZAHL - 10)

VORWÄRTS :lZweig

Baum ((:lZweig * 1) / 3) KAP XKO YKO

]

RÜCKKEHR

ENDE

```

// Tippen in die Befehllinie (Arbeitsfenster):
Pappeln
//
UM Schnecke
BILD STIFTAB VERSTECKTIGEL
BESCHLEUNIGE STIFTDICKE 3 FARBEIGEL SCHWARZ
SETZE "Zähler 1
// Vorzeichnung der Schale. //
WIEDERHOLE 550 [
VORWÄRTS 0,01 * : Zähler RECHTS 3 SETZE "Zähler : Zähler +1
]
LINKS 120 VORWÄRTS 70 // man färbt den Körper //
LINKS 20 VORWÄRTS 20
WIEDERHOLE 3 [
RECHTS 30 VORWÄRTS 9
RECHTS 29 VORWÄRTS 10
]
RECHTS 20 VORWÄRTS 170
STIFTHOCH
RÜCKWÄRTS 175
RECHTS 90 VORWÄRTS 37
STIFTAB
// Vorzeichnung der Antennen. //
RECHTS 30 VORWÄRTS 30 RÜCKWÄRTS 30
RECHTS 15 VORWÄRTS 30
STIFTHOCH GEHE [ 86 (-64) ] STIFTAB GEHE [57 (-64) ]
FÜLLFARBE [ -21 (-10)] LINDE //man färbt die Schale. //
FÜLLFARBE [ 50 (-80)] ORANGE // man färbt den Körper. //
STIFTDICKE 1
ENDE

```

```

// Tippen in die Befehllinie (Arbeitsfenster):
Schnecke
//

```

```

UM Matriz
VERSTECKTIGEL BILD
SETZE "y 0
WIEDERHOLE 50
[
SETZE "x 0
WIEDERHOLE 50
[
PUNKT [ :y :x ]
SETZE "x :x + 5
]
SETZE "y :y + 5
]
ENDE

```

```

// Tippen in die Befehllinie (Arbeitsfenster):
Matriz
//

```

```

UM Kreis
SETZE "posi LAGE
ELLIPSE [ XKO YKO 35 35 ]
BRINGEZURÜCK :posi
ENDE
UM Torus
BILD BESCHLEUNIGE STIFTHOCH VERSTECKTIGEL
WIEDERHOLE 180 [ Kreis VORWÄRTS 4 RECHTS 2]
ENDE

```

```

// Tippen in die Befehllinie (Arbeitsfenster):
Torus
//

```

```

// Wenn die Differentialgleichungen beginnen zu zeichnen!
Hier ein "langames" Verfahren. //

```

```

UM Kurve1 :n
STARTEZUFALL -280 280
BILD LÖSCHTEXT VERSTECKTIGEL
TEIL [ "x "y ]
SETZE "x [] SETZE "y []
Poly 150 (360 / :n)
SETZE "h 0,05
WIEDERHOLE 100 [ // Nur 100 Wiederholungen. //
LÖSCHTEXT SCHREIBLINIE WHZAHL
FARBEIGEL REST WHZAHL 17
SETZE "xxm [] SETZE "yym []
WIEDERHOLE :n [ // die vorhergehende Kurve verfolgt das folgende.//
SETZE "mg ( REST WHZAHL :n ) + 1
SETZE "xxm SATZ :xxm ( ELEMENT :x WHZAHL ) + ( :h * (( ELEMENT :x :mg ) - ELEMENT :x WHZAHL ))
SETZE "yym SATZ :yym ( ELEMENT :y WHZAHL ) + ( :h * (( ELEMENT :y :mg ) - ELEMENT :y WHZAHL ))
SEGMENT [ ELEMENT :x :mg ELEMENT :y :mg ] [ ELEMENT :xxm WHZAHL ELEMENT :yym WHZAHL ]
// Man zieht hier. //
]
WIEDERHOLE :n [ // Man aktualisiert die Liste der neuen Koordinaten.//
ERSETZ :x ( ELEMENT : xxm WHZAHL ) [ WHZAHL ]
ERSETZ :y ( ELEMENT : yym WHZAHL ) [ WHZAHL ]
]
]
ENDE

```

// und ein "schnelles" Verfahren. //

```

UM Kurve2 :n
STARTEZUFALL -280 280
BILD LÖSCHTEXT VERSTECKTIGEL
TEIL [ "x "y ]
SETZE "x [] SETZE "y []
Poly 150 (360 / :n)
SETZE "h 0,05
WIEDERHOLE 300 [ // 300 Wiederholungen. //
LÖSCHTEXT SCHREIBLINIE WHZAHL
FARBEIGEL REST WHZAHL 17
SETZE "xxm [] SETZE "yym []
// Man berechnet die neuen Koordinaten der extremen Punkte der Listen :x und :y um Berechnungen
zu vereinfachen. Man kann jetzt diese zwei Listen je nach der Wiederholungen abstumpfen, von wo
Beschleunigung der Vorzeichnung. //
SETZE "xd ( LETZTES :x ) + ( :h * (( ERSTES :x ) - ( LETZTES :x ) ) )
SETZE "yd ( LETZTES :y ) + ( :h * (( ERSTES :y ) - ( LETZTES :y ) ) )
SETZE "xxm SATZ :xxm ( ERSTES :x ) + ( :h * (( ELEMENT :x 2 ) - ( ERSTES :x ) ) )
SETZE "yym SATZ :yym ( ERSTES :y ) + ( :h * (( ELEMENT :y 2 ) - ( ERSTES :y ) ) )
SEGMENT [ ELEMENT :x 2 ELEMENT :y 2 ] [ LETZTES :xxm LETZTES :yym ]
SEGMENT [ :xd :yd ] [ LETZTES :xxm LETZTES :yym ]
SETZE "x OHNEERSTES :x SETZE "y OHNEERSTES :y
WIEDERHOLE ( :n - 2 ) [ // die vorhergehende Kurve verfolgt das folgende.//
SETZE "xxm SATZ :xxm ( ERSTES :x ) + ( :h * (( ELEMENT :x 2 ) - ( ERSTES :x ) ) )
SETZE "yym SATZ :yym ( ERSTES :y ) + ( :h * (( ELEMENT :y 2 ) - ( ERSTES :y ) ) )
SEGMENT [ ELEMENT :x 2 ELEMENT :y 2 ] [ LETZTES :xxm LETZTES :yym ] // Man zieht hier. //
SETZE "x OHNEERSTES :x SETZE "y OHNEERSTES :y
]
SETZE "xxm SATZ :xxm :xd SETZE "yym SATZ :yym :yd
SETZE "x [] SETZE "y []
WIEDERHOLE :n [ // Man aktualisiert die Liste der neuen Koordinaten.//
SETZE "x SATZ :x ERSTES :xxm
SETZE "y SATZ :y ERSTES :yym
SETZE "xxm OHNEERSTES :xxm
SETZE "yym OHNEERSTES :yym
]
]
ENDE

```

```

UM Poly :Seite :Winkel
SCHREIBLINIE [ Berechnung der Anfangspunkte. ]
ROLLAUF STIFTHOCH RÜCKWÄRTS 100 RECHTS 90
RÜCKWÄRTS 50 LINKS 90

```

```
STIFTHOCH
WIEDERHOLE GANZZAHL ( 360 / :Winkel )
[
VORWÄRTS :Seite
RECHTS :Winkel
SETZE "x SATZ :x XKO
SETZE "y SATZ :y YKO
]
ENDE
```

Kurve1 5

// Tippen in der Befehllinie (Arbeitsfenster):

Kurve1 5

oder

Kurve2 5

(die Vorzeichnung von Kurve1 ist langsamer als jene von Kurve2)

//